



Theorembeweiserpraktikum – SS 2016

<http://pp.ipd.kit.edu/lehre/SS2016/tba>

Lösung 1: Deduktion

Abgabe: 25. April 2016

Besprechung: 26. April 2016

1 Natürliches Schließen

In dieser Aufgabe geht es um den Kalkül des natürlichen Schließens, mit dessen Hilfe einige Lemmas der Aussagen-Logik bewiesen werden sollen (nächste Seite).

Für die Beweise gelten die folgenden Spielregeln:

- Es dürfen nur die Befehle **proof**, **assume**, **have**, **show**, **next**, **qed**, und **from** verwendet werden, sowie darauf aufbauende Abkürzungen wie **then**, **with**, **..** und **..**.
- Der Befehl **proof** darf nur als **proof** (*rule regel*) verwendet werden, wobei die Regel eine der folgenden ist: (Anzeigen der Lemmas mittel **thm lemma-Name**)

$impI: (A \implies B) \implies A \longrightarrow B,$ $impE: A \longrightarrow B \implies A \implies (B \implies C) \implies C,$
 $conjI: A \implies B \implies A \wedge B,$ $conjE: A \wedge B \implies (A \implies B \implies C) \implies C,$
 $disjI1: A \implies A \vee B,$ $disjE:$
 $disjI2: B \implies A \vee B,$ $A \vee B \implies (A \implies C) \implies (B \implies C) \implies C,$
 $notI: (A \implies False) \implies \neg A,$ $notE: \neg A \implies A \implies B,$
 $iffI: (A \implies B) \implies (B \implies A) \implies A \longleftrightarrow B,$
 $iffE: A \longleftrightarrow B \implies (A \longrightarrow B \implies B \longrightarrow A \implies C) \implies C,$
 $ccontr: (\neg A \implies False) \implies A$
 $classical: (\neg A \implies A) \implies A$

Alle diese Regeln, außer den letzten beiden, sind als Standard-Regeln vorgegeben, das heißt der Befehl **proof** (*rule*) (oder kurz **proof**) wählt die passende Regel aus, auch ohne dass man sie explizit angibt. Lassen Sie nur Namen von Regeln weg, die sie zuvor zumindest einmal explizit verwendet haben.

Beispiel

lemma *imp_uncurry*: " $(P \longrightarrow (Q \longrightarrow R)) \longrightarrow P \wedge Q \longrightarrow R$ "

proof(*rule impI*)

assume *PQR*: " $P \longrightarrow (Q \longrightarrow R)$ "

show " $P \wedge Q \longrightarrow R$ "

proof — Das (*rule impI*) kann weglassen werden

assume " $P \wedge Q$ "

then have " P " **by**(*rule conjE*)

with *PQR*

have " $Q \longrightarrow R$ " **by**(*rule impE*)

```

from  $\langle P \wedge Q \rangle$ 
have "Q".. — Hier steht eigentlich by(rule conjE)
with  $\langle Q \longrightarrow R \rangle$ 
show R..
qed
qed

```

```

lemma I: "A  $\longrightarrow$  A"
by (rule impI)

```

```

lemma "A  $\wedge$  B  $\longrightarrow$  B  $\wedge$  A"
proof
  assume "A  $\wedge$  B"
  then have "A" by(rule conjE)

```

```

from  $\langle A \wedge B \rangle$ 
have "B"..

```

```

from  $\langle B \rangle \langle A \rangle$ 
show "B  $\wedge$  A" by (rule conjI)
qed

```

```

lemma "A  $\wedge$  B  $\longrightarrow$  A  $\vee$  B"
proof
  assume "A  $\wedge$  B"
  then have "A"..
  then show "A  $\vee$  B" by (rule disjI1)
qed

```

```

lemma " $((A \vee B) \vee C) \longrightarrow A \vee (B \vee C)$ "
proof
  assume " $(A \vee B) \vee C$ "
  then show "A  $\vee$  (B  $\vee$  C)"
  proof(rule disjE)
    assume "A  $\vee$  B"
    then show "A  $\vee$  (B  $\vee$  C)"
    proof
      assume A
      then show "A  $\vee$  (B  $\vee$  C)" by (rule disjI1)
    next
      assume B
      then have "B  $\vee$  C"..
      then show "A  $\vee$  (B  $\vee$  C)" by (rule disjI2)
    qed
  next
    assume C
    then have "B  $\vee$  C"..
    then show "A  $\vee$  (B  $\vee$  C)"..
  qed
qed

```

```

lemma K: "A  $\longrightarrow$  B  $\longrightarrow$  A"

```

```

proof
  assume "A"
  show "B  $\rightarrow$  A"
  proof
    from  $\langle A \rangle$ 
    show "A"..
  qed
qed

```

lemma " $A \vee A \leftrightarrow A \wedge A$ "

```

proof
  assume "A  $\vee$  A"
  then show "A  $\wedge$  A"
  proof
    assume A
    from  $\langle A \rangle$  and  $\langle A \rangle$ 
    show "A  $\wedge$  A"..
  next
    assume A
    from  $\langle A \rangle$  and  $\langle A \rangle$ 
    show "A  $\wedge$  A"..
  qed
next
  assume "A  $\wedge$  A"
  then have "A"..
  then show "A  $\vee$  A"..
qed

```

lemma S : " $(A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C$ "

```

proof
  assume ABC: "A  $\rightarrow$  B  $\rightarrow$  C"
  show " $(A \rightarrow B) \rightarrow A \rightarrow C$ "
  proof
    assume "A  $\rightarrow$  B"
    show "A  $\rightarrow$  C"
    proof
      assume "A"
      with  $\langle A \rightarrow B \rangle$ 
      have "B" by (rule impE)

      from ABC and  $\langle A \rangle$ 
      have "B  $\rightarrow$  C"..

      from  $\langle B \rightarrow C \rangle$  and  $\langle B \rangle$ 
      show "C"..
    qed
  qed
qed

```

lemma " $(A \rightarrow B) \rightarrow (B \rightarrow C) \rightarrow A \rightarrow C$ "

```

proof
  assume "A  $\rightarrow$  B"

```

```

show "(B  $\longrightarrow$  C)  $\longrightarrow$  A  $\longrightarrow$  C"
proof
  assume "B  $\longrightarrow$  C"
  show "A  $\longrightarrow$  C"
  proof
    assume "A"
    with ⟨A  $\longrightarrow$  B⟩
    have B ..
    with ⟨B  $\longrightarrow$  C⟩
    show C ..
  qed
qed
qed

lemma " $\neg \neg$  A  $\longrightarrow$  A"
proof
  assume " $\neg \neg$  A"
  show A
  proof(rule ccontr)
    assume " $\neg$ A"
    with ⟨ $\neg \neg$  A⟩
    show False by (rule notE)
  qed
qed

lemma "A  $\longrightarrow$   $\neg \neg$  A"
proof
  assume A
  show " $\neg \neg$  A"
  proof(rule notI)
    assume " $\neg$  A"
    from this ⟨A⟩
    show False ..
  qed
qed

lemma " $(\neg$  A  $\longrightarrow$  B)  $\longrightarrow$  ( $\neg$  B  $\longrightarrow$  A)"
proof
  assume " $\neg$  A  $\longrightarrow$  B"
  show " $\neg$  B  $\longrightarrow$  A"
  proof
    assume " $\neg$  B"
    show A
    proof(rule ccontr)
      assume " $\neg$ A"
      with ⟨ $\neg$  A  $\longrightarrow$  B⟩
      have "B" ..
      with ⟨ $\neg$  B⟩
      show False ..
    qed
  qed
qed

```

```

lemma " $((A \rightarrow B) \rightarrow A) \rightarrow A$ "
proof
  assume ABA: " $((A \rightarrow B) \rightarrow A)$ "
  show A
  proof(rule classical)
    assume " $\neg A$ "

    have " $A \rightarrow B$ "
    proof
      assume A
      with  $\langle \neg A \rangle$ 
      show B..
    qed
    with ABA
    show A..
  qed
qed

lemma " $A \vee \neg A$ "
proof(rule classical)
  assume " $\neg (A \vee \neg A)$ "

  have " $\neg A$ "
  proof
    assume A
    then have " $A \vee \neg A$ " ..
    with  $\langle \neg (A \vee \neg A) \rangle$ 
    show False..
  qed
  then show " $A \vee \neg A$ "..
qed

lemma deMorgan1: " $\neg (A \vee B) \longleftrightarrow \neg A \wedge \neg B$ "
proof(rule iffI)
  assume " $\neg (A \vee B)$ "
  show " $\neg A \wedge \neg B$ "
  proof
    show " $\neg A$ "
    proof
      assume "A"
      then have " $A \vee B$ "..
      with  $\langle \neg (A \vee B) \rangle$ 
      show False ..
    qed
  next
    show " $\neg B$ "
    proof
      assume "B"
      then have " $A \vee B$ "..
      with  $\langle \neg (A \vee B) \rangle$ 
      show False ..
    qed
  qed

```

```

qed
next
  assume " $\neg A \wedge \neg B$ "
  then have " $\neg A$ "..

  from  $\langle \neg A \wedge \neg B \rangle$ 
  have " $\neg B$ "..

  show " $\neg (A \vee B)$ "
  proof
    assume " $A \vee B$ "
    then show False
    proof
      assume "A"
      with  $\langle \neg A \rangle$ 
      show False..
    next
      assume "B"
      with  $\langle \neg B \rangle$ 
      show False..
    qed
  qed
qed
qed

lemma deMorgan2: " $\neg (A \wedge B) \longleftrightarrow \neg A \vee \neg B$ "
proof
  assume " $\neg (A \wedge B)$ "
  show " $\neg A \vee \neg B$ "
  proof(rule classical)
    assume " $\neg (\neg A \vee \neg B)$ "
    have "A"
    proof (rule ccontr)
      assume " $\neg A$ "
      then have " $\neg A \vee \neg B$ "..
      with  $\langle \neg (\neg A \vee \neg B) \rangle$ 
      show False..
    qed

    have "B"
    proof (rule ccontr)
      assume " $\neg B$ "
      then have " $\neg A \vee \neg B$ "..
      with  $\langle \neg (\neg A \vee \neg B) \rangle$ 
      show False..
    qed

    from  $\langle A \rangle \langle B \rangle$  have " $A \wedge B$ "..
    with  $\langle \neg (A \wedge B) \rangle$ 
    show ?thesis..
  qed
next
  assume " $\neg A \vee \neg B$ "

```

```

then show " $\neg (A \wedge B)$ "
proof
  assume " $\neg A$ "
  show ?thesis
  proof
    assume " $A \wedge B$ "
    then have "A"..
    with  $\langle \neg A \rangle$ 
    show False..
  qed
next
  assume " $\neg B$ "
  show ?thesis
  proof
    assume " $A \wedge B$ "
    then have "B"..
    with  $\langle \neg B \rangle$ 
    show False..
  qed
qed
qed

```

Anmerkung: Ist Ihnen bei den Beweisen der De Morgan-Regeln etwas aufgefallen?

deMorgan1 kommt ohne Regeln der klassischen Logik aus, für *deMorgan2* braucht man jedoch Fallunterscheidung. D.h. $\neg (A \wedge B) \longrightarrow \neg A \vee \neg B$ ist in intuitionistischer Logik nicht herleitbar. Dies gilt übrigens auch für die Theoreme $(\neg A \longrightarrow B) \longrightarrow \neg B \longrightarrow A$, $((A \longrightarrow B) \longrightarrow A) \longrightarrow A$, $A \vee \neg A$ und $\neg \neg A \longrightarrow A$